

Pràctiques integrades

Pràctica 2

Matemàtiques, curs 2001-2002.

2 Càlculs Algebraics

Maple és un “C.A.S” , i.e. un “**Computer Algebra System**” (en català es diu normalment que és un ***Manipulador Algebraic***). Això significa que Maple coneix totes les regles algebraiques que vosaltres sabeu. Al mateix temps que aneu progressant en el Càlcul, les Equacions diferencials i l'Àlgebra lineal veureu que Maple també conté les operacions essencials d'aquests temes introduïdes en el seu gran conjunt de comandes.

En aquesta secció aprendreu com introduir una expressió algebraica i a donar valors a les seves variables. Després aprendreu les comandes que us permetran expandir, factoritzar i simplificar expressions.

```
> restart;
```

2.1 La comanda subs()

Exemple 1:

Com a primer exemple comencem amb l'expressió $3x^2 + 8$ i assignem-li com a nom W.

```
> W:=3*x^2+8;
```

Ara suposem que volem substituir la variable x de l'expressió $3x^2 + 8$ pel valor 4. La forma més ràpida de fer-ho és utilitzar la comanda de Maple subs(). Aquí veiem com

```
> subs(x=4,3*x^2+8);
```

De forma alternativa podem aplicar la comanda subs() a W.

```
> subs(x=4,W);
```

Exemple 2:

La comanda subs() també funciona bé amb valors simbòlics:

Per a substituir x per $5 + 2u$ en l'expressió $3x^2 + 8$ executem la línia següent (En aquest cas posem l'etiqueta M al resultat).

```
> W:=3*x^2+8;
```

```
> M:=subs(x=5+2*u,W);
```

I ara per a fer que Maple “multipliqui” aquesta expressió utilitzem la comanda expand().

```
> expand(M);
```

Exemple 3:

La comanda subs() és molt versàtil. Podem utilitzar-la per a avaluar expressions en les que intervenen més d'una variable. Aquí substituïm la x per 7 i la y per 12 en l'expressió $U = \frac{2x^2}{5} + 3y$.

```
> U:=(2/5)*x^2+3*y;
```

```
> subs(x=7,y=12,U);
```

```
> evalf(%);
```

Exemple 4:

També podem utilitzar la comanda `subs( )` per a substituir un valor en una equació. Això és el tipus de coses que cal fer per a verificar si un determinat valor “satisfà” l’equació. En les línies que venen a continuació substituïrem diferents valors en l’equació $x^3 - 5x^2 + 7x - 12 = 0$. És algun d’aquests valors una solució de l’equació?

Noteu que utilitzem “:=” per assignar el nom i només “=” per a l’equació mateixa.

```
> eqn:=x^3-5*x^2+7*x-12=0;
> subs(x=3,eqn);
> subs(x=4,eqn);
> subs(x=5,eqn);
```

Exercici 2.1

Assigneu el nom k a l’expressió $x^2 + 4x - 3$. Un cop fet això dieu M a l’expressió $k^2 - 9$. Finalment feu que Maple calculi $3M + 6$. Nota: per fer que Maple multipliqui les expressions utilitzeu la comanda `expand( )`.

És a dir, introduïu

```
> expand(3*M+6);
```

Apreneu més coses sobre la comanda `expand( )` en l’apartat següent.

Exercici 2.2

Desenvolupeu $(1 + x)^4$ utilitzant la comanda `expand( )`.

Exercici 2.3

Sigui $P = ax^3 + bx^2 + cx + d$. Calculeu P si $x = 0.01$, $a = -\frac{1}{5}$, $b = \frac{2}{5}$, $c = 0$, i $d = \frac{13}{15}$.

Exercici 2.4

Utilitzeu la comanda `subs( )` per a verificar si algun dels nombres 1, 2 o 3 és una solució de l'equació:
 $x^3 - 16x^2 + 51x - 36 = 0$

2.2 La comanda `expand( )`

L'ús principal de la comanda `expand( )` és el de “fer les multiplicacions” en els productes d'expressions polinòmiques. També es pot utilitzar per a desenvolupar funcions trigonomètriques i altres tipus de funcions més generals.

Exemple 1:

Utilitzeu la comanda `expand( )` per efectuar les multiplicacions en $(x + 2)^2 (3x - 3)(x + 5)$.

```
> k:=(x+2)^2*(3*x-3)*(x+5);  
> expand(k);
```

Exemple 2:

Maple aplica algunes de les identitats trigonomètriques més comuns per a desenvolupar $\sin(2x)$ i $\cos(2x)$.

```
> expand(sin(2*x));  
> expand(cos(2*x));
```

Feu alguna prova desenvolupant el sinus i el cosinus d'alguns altres múltiples enters de x . Per exemple: $\sin(3x)$, $\cos(6x)$, etc.

Exemple 3:

Aquí hi ha un exemple final. Feu que Maple multipliqui l'expressió $x^{(\frac{1}{2})} (x^{(\frac{3}{2})} + x^{(-\frac{1}{2})})$

```
> h:=x^(1/2)*(x^(3/2)+x^(-1/2));  
> expand(h);
```

Exercici 2.6

Desenvolueu $(x + 1)^n$ per a $n = 2, 3$ i 4 .

2.3 La comanda `factor( )`

Exemple 1:

Factoritzem l'expressió: $3x^2 - 10x - 8$

```
> w:=3*x^2-10*x-8;
> factor(w);
```

O podem fer el mateix en una sola línia:

```
> factor(3*x^2-10*x-8);
```

Exemple 2:

Primer desenvolupem l'expressió $2(x-2)(2x^2+5x+2)(x+4)$. Després apliquem la comanda `factor( )` al resultat. Podeu explicar per què el resultat final sembla diferent de l'expressió original?

```
> H:=2*(x-2)*(2*x^2+5*x+2)*(x+4);
> ans:=expand(H);
> factor(ans);
```

Exemple 3:

Maple pot factoritzar expressions amb més d'una variable.

Factoritzem l'expressió: $x^2y + 2xy + y$

```
> h:=x^2*y+2*x*y+y;
> factor(h);
```

Exemple 4:

Si Maple no pot factoritzar una expressió utilitzant nombres racionals (és a dir, enters i fraccions) donarà com a resultat el mateix que heu introduït sense cap canvi.

```
> factor(3*x^2-10*x-9);
```

Exemple 5:

La comanda `factor( )` no està limitada als polinomis. Es pot utilitzar per a factoritzar altres formes.

Factoritzeu $(\sin x)^2 - (\cos x)^2$.

```
> factor((sin(x))^2-(cos(x))^2);
```

Exemple 6:

Si la comanda `factor( )` s'utilitza amb una expressió racional, es factoritza el numerador i el denominador i els factors comuns es cancel·len per a simplificar l'expressió:

```
> A:=(x^3-7*x^2+15*x-9)/(x^2+4*x+4);
> factor(A);
> B:=(x^3-7*x^2+15*x-9)/(x^2-4*x+3);
> factor(B);
```

El següent exemple us permet veure la forma factoritzada sense les simplificacions.

Exemple 7:

Les comandes de Maple `numer( )` i `denom( )` us permeten aïllar tant el numerador com el denominador d'una fracció. Aquí utilitzem aquestes comandes per examinar els factors del numerador i el denominador per separat (és a dir, abans de les simplificacions dels factors comuns).

```
> B:=(x^3-7*x^2+15*x-9)/(x^2-4*x+3);
> factor(numer(B)); factor(denom(B));
```

Exercici 2.8

Factoritzeu l'expressió $3x^4 - 2x^3 + 22x^2 - 18x - 45$.

Exercici 2.9

Factoritzeu l'expressió $x^{(\frac{1}{2})} - x^{(\frac{3}{2})}$ i després utilitzeu la comanda `expand( )` per a comprovar el resultat.

2.4 La comanda `simplify( )`**Exemple 1:**

Considereu l'expressió $(\cos x)^5 + (\sin x)^4 + 2(\cos x)^2 - 2(\sin x)^2 - \cos(2x)$. Maple pot aplicar identitats per a simplificar expressions matemàtiques llargues, com ara expressions trigonomètriques.

```
> V:=cos(x)^5 + sin(x)^4 + 2*cos(x)^2 - 2*sin(x)^2 - cos(2*x);
> simplify(V);
```

Exemple 2:

Les expressions trigonomètriques amb arguments donats com a múltiples d'algun angle queden simplificades en termes de funcions trigonomètriques d'aquest angle si és possible:

```
> simplify(sin(5*t)+sin(3*t));
```

Exemple 3:

La comanda `simplify( )` es pot utilitzar per a sumar expressions racionals. A continuació reescrivim la suma $\frac{1}{x+1} + \frac{x}{x-1}$ com una única fracció.

```
> M:=(1/(x+1))+(x/(x-1));
> simplify(M);
```

Exercici 2.11:

Simplifiqueu l'expressió $\frac{7}{x+2} + \frac{3x}{(x+2)^2}$

Exercici 2.12

Com simplifica Maple $\sin(3t) - \sin(7t)$? Si aquesta expressió “simplificada” és o no és útil dependrà del que es tingui planejat fer amb ella.